

```

359 POKE 1024+1+40*PY,48
360 FOR I=0 TO 39
361 LET PX=I: LET PY=INT((TM(I)-MX)/(MN-MX)*24+0.5)
362 POKE 55296+PX+40*PY,BZ
363 POKE 1024+PX+40*PY,160
364 NEXT I

```

Na horní a dolní hranu obrazovky se zobrazí meze oboru hodnot.

$\langle \text{Graf funkce} \rangle + \equiv$

```

365 PRINT CHR$(19);BN$;MX;
366 LET PX=0: LET PY=24: GOSUB 131
367 PRINT MN;

```

Po zobrazení grafu se čeká na stisk klávesy a pak se vrací na volbu metody.

$\langle \text{Graf funkce} \rangle + \equiv$

```

368 GET K$: IF K$="" THEN 368
369 GOTO 294

```

3.2.4 Bisekce

Na začátku bisekce uživatel zadá krajní body do proměnných A, B.

$\langle \text{Bisekce} \rangle \equiv$

```

370 LET RA=15: GOSUB 139: INPUT A
371 LET RA=16: GOSUB 139: INPUT B

```

Ze slušnosti je zařízeno, aby platilo $A < B$.

$\langle \text{Bisekce} \rangle + \equiv$

```

372 IF B<A THEN LET C=A: LET A=B: LET B=C

```

Pokud nejsou v krajních bodech různá znaménka, vypíše se chybové hlášení a metoda se ukončí.

$\langle \text{Bisekce} \rangle + \equiv$

```

373 LET X=A: GOSUB 173: LET D=F
374 LET X=B: GOSUB 173
375 IF SGN(F)*SGN(D)<0 THEN 378
376 LET RA=17: GOSUB 158
377 GOTO 390

```

V tuto chvíli již jsou splněny předpoklady a lze provést iterace bisekce.

$\langle \text{Bisekce} \rangle + \equiv$

```

378 LET RA=18: GOSUB 139: INPUT N
379 FOR I=1 TO N
380 LET C=(A+B)/2
381 LET X=A: GOSUB 173: LET D=F
382 LET X=C: GOSUB 173
383 IF SGN(F)*SGN(D)<0 THEN 385
384 LET A=C: GOTO 386
385 LET B=C
386 PRINT I;": ";A,B
387 NEXT I
388 LET C=(A+B)/2
389 PRINT BZ$;C;BN$

```

Po skončení bisekce je třeba skočit na společný konec metod.

$\langle \text{Bisekce} \rangle + \equiv$

```

390 GOTO 317

```