

1076. The global variable *adjust_tail* will be non-null if and only if the current box might include adjustments that should be appended to the current vertical list.

⟨Append box *cur_box* to the current list, shifted by *box_context* 1076⟩ ≡

```

begin if cur_box ≠ null then
  begin shift_amount(cur_box) ← box_context;
  if abs(mode) = vmode then
    begin append_to_vlist(cur_box);
    if adjust_tail ≠ null then
      begin if adjust_head ≠ adjust_tail then
        begin link(tail) ← link(adjust_head); tail ← adjust_tail;
        end;
        adjust_tail ← null;
      end;
    if mode > 0 then build_page;
  end
else begin if abs(mode) = hmode then space_factor ← 1000
  else begin p ← new_noad; math_type(nucleus(p)) ← sub_box; info(nucleus(p)) ← cur_box;
    cur_box ← p;
  end;
  link(tail) ← cur_box; tail ← cur_box;
end;
end;
end

```

This code is used in section 1075.

1077. ⟨Store *cur_box* in a box register 1077⟩ ≡

```

if box_context < box_flag + 256 then eq_define(box_base − box_flag + box_context, box_ref, cur_box)
else geq_define(box_base − box_flag − 256 + box_context, box_ref, cur_box)

```

This code is used in section 1075.

1078. ⟨Append a new leader node that uses *cur_box* 1078⟩ ≡

```

begin ⟨Get the next non-blank non-relax non-call token 404⟩;
if ((cur_cmd = hskip) ∧ (abs(mode) ≠ vmode)) ∨ ((cur_cmd = vskip) ∧ (abs(mode) = vmode)) then
  begin append_glue; subtype(tail) ← box_context − (leader_flag − a_leaders);
  leader_ptr(tail) ← cur_box;
  end
else begin print_err("Leaders not followed by proper glue");
  help3("You should say \"\leaders<box or rule><hskip or vskip>\".")
  ("I found the <box or rule>, but there's no suitable")
  ("<hskip or vskip>, so I'm ignoring these leaders."); back_error; flush_node_list(cur_box);
end;
end

```

This code is used in section 1075.